Application Demonstration
www.maplesoft.com

Trajectory of a Bullet Using an Air Resistance Correlation

Introduction

This application models the trajectory of a bullet fired from a gun. Specifically, the application numerically solves the differential equations that describe the motion of the bullet (assuming two degrees of freedom), together with a correlation for air resistance. The predictions are compared to manufacturer-supplied velocity-range data for a specific type of bullet; the model closely matches the manufacturer's data.

After leaving the muzzle, a bullet travels under the influence of air resistance. This is described by a drag coefficient, which is a function of the bullet velocity (or Mach number).

Typically, the G model is used to compute the drag coefficient for bullets; this tabulates experimentally-determined drag coefficients and Mach numbers for several standard projectile shapes. For a specific bullet, the drag coefficient is multiplied by a form factor, which describes an individual bullet's variation from the standard design.

There are several variants of the G model, ranging from G1 to G7. Typically, bullet manufacturers recommend the preferred variant, and provide the value of the form factor for each of their bullets. This application uses data for a bullet for which the G7 model is recommended.

```
> restart :
plot_style := labeldirections = [horizontal, vertical], thickness = 3, color = "DarkGreen", axes
    = boxed, size = [800, 400], titlefont = [Calibri, 16], axesfont = [Calibri], labelfont = [Calibri,
    12], background = ColorTools:-Color( "RGB", [218 / 255, 223 / 255, 225 / 255]) , axis = [gridlines
    = [10, color = white]], legendstyle = [font = [Calibri]] :
```

Differential Equations for the Trajectory of a Bullet

A bullet leaves a muzzle at an initial angle and velocity of θ_0 and v_0. The bullet travels is considered a point-mass, and travels under the influence of gravity and aerodynamic drag.



Drag force on bullet

$$> Fd := -\frac{\rho \cdot v(t)^2}{2} \cdot A \cdot i \cdot Cd :$$

Gravitational force on bullet

$$> Fg := -m \cdot g \sin(\theta(t)) :$$

Force balance on bullet

$$> de1 := m \cdot \frac{d}{dt} v(t) = Fg + Fd :$$

Rate of change of angle

$$> de2 := \frac{d}{dt} \theta(t) = -\frac{g \cdot \cos(\theta(t))}{v(t)} :$$

Rate of change of horizontal and vertical position

$$> de3 := \frac{d}{dt} x(t) = v(t) \cdot \cos(\theta(t)) :$$

$$> de4 := \frac{d}{dt} y(t) = v(t) \cdot \sin(\theta(t)) :$$

Initial conditions

$$> ic := v(0) = v_0, \theta(0) = \theta_0, x(0) = 0, y(0) = 2 :$$

where

- t – time (s)
- v(t) – instantaneous velocity (m s^{-1})
- x – horizontal travel (m)
- y – vertical travel (m)

- θ – angle of the velocity relative to the horizontal (radians)
- m – mass of bullet(kg)
- A – cross-sectional area of bullet (m^2)
- Cd – drag coefficient
- i – form factor of bullet
- ρ – air density (kg m^{-3})
- g – acceleration due to gravity (m s^{-2})

Parameters

The following bullet diameter, mass, G7 form factor and muzzle velocity were found on the 308 Winchester 185gr Juggernaut Target bullet web page.

Bullet diameter

> $d := convert(0.308, units, 'inches', 'meter')$

$$d := 0.007823200000 \quad (3.1)$$

Mass of bullet

> $m := convert(185., units, 'grain', 'kg')$

$$m := 0.01198779835 \quad (3.2)$$

Form factor of bullet

> $i := 0.98 :$

Muzzle velocity

> $v_0 := convert(2628., units, 'ft/s', 'm/s')$

$$v_0 := 801.0144000 \quad (3.3)$$

Angle of discharge

> $\theta_0 := 0 :$

Cross -sectional area of bullet

> $A := \frac{\pi d^2}{4} :$

Air density at sea level

> $\rho := 1.225 :$

Speed of sound at sea level

> $a := 340.3 :$

Gravity

> $g := 9.8$:

G7 standard bullet drag coefficient (found here)

> $G7\_Mach\_number := [0., 0.05, 0.10, 0.15, 0.20, 0.25, 0.30, 0.35, 0.40, 0.45, 0.50, 0.55, 0.60, 0.65, 0.70, 0.725, 0.75, 0.775, 0.80, 0.825, 0.85, 0.875, 0.90, 0.925, 0.95, 0.975, 1.0, 1.025, 1.05, 1.075, 1.10, 1.125, 1.15, 1.20, 1.25, 1.30, 1.35, 1.40, 1.50, 1.55, 1.60, 1.65, 1.70, 1.75, 1.80, 1.85, 1.90, 1.95, 2.00, 2.05, 2.10, 2.15, 2.20, 2.25, 2.30, 2.35, 2.40, 2.45, 2.50, 2.55, 2.60, 2.65, 2.70, 2.75, 2.80, 2.85, 2.90, 2.95, 3.00, 3.10, 3.20, 3.30, 3.40, 3.50, 3.60, 3.70, 3.80, 3.90, 4.00, 4.20, 4.40, 4.60, 4.80, 5.00]$:

> $G7\_drag\_coefficient := [0.1198, 0.1197, 0.1196, 0.1194, 0.1193, 0.1194, 0.1194, 0.1194, 0.1193, 0.1193, 0.1194, 0.1193, 0.1194, 0.1197, 0.1202, 0.1207, 0.1215, 0.1226, 0.1242, 0.1266, 0.1306, 0.1368, 0.1464, 0.1660, 0.2054, 0.2993, 0.3803, 0.4015, 0.4043, 0.4034, 0.4014, 0.3987, 0.3955, 0.3884, 0.3810, 0.3732, 0.3657, 0.3580, 0.3440, 0.3376, 0.3315, 0.3260, 0.3209, 0.3160, 0.3117, 0.3078, 0.3042, 0.3010, 0.2980, 0.2951, 0.2922, 0.2892, 0.2864, 0.2835, 0.2807, 0.2779, 0.2752, 0.2725, 0.2697, 0.2670, 0.2643, 0.2615, 0.2588, 0.2561, 0.2533, 0.2506, 0.2479, 0.2451, 0.2424, 0.2368, 0.2313, 0.2258, 0.2205, 0.2154, 0.2106, 0.2060, 0.2017, 0.1975, 0.1935, 0.1861, 0.1793, 0.1730, 0.1672, 0.1618]$:

The G7 drag coefficient data should be fitted to a spline so that it can be used in the theoretical model

> $Cd := CurveFitting\text{:-}Spline(G7\_Mach\_number, G7\_drag\_coefficient, M)$:
$plot(Cd, M = 0..5, labels = [\text{"Mach Number"}, \text{"Drag Coefficient"}], title = \text{"Drag Coefficient for the G7 Standard Projectile"}, plot\_style)$



Replace the Mach number with the bullet velocity and the speed of sound

> $Cd := subs\left(M = \frac{v(t)}{a}, Cd\right)$:

Numerical Solution of Differential Equations

> $res := dsolve(\{de1, de2, de3, de4, ic\}, \{\theta(t), v(t), x(t), y(t)\}, numeric)$

$res := \mathbf{proc}(x\_rkf45)\ \ldots\ \mathbf{end\ proc}$ **(4.1)**

Trajectory of bullet through space

```
> plots:-odeplot(res, [x(t), y(t)], t = 0 ..0.55, axes = boxed, labels = ["Horizontal Distance (m)",
    "Vertical Distance (m)"], title = "Trajectory of Bullet", plot_style)
```



Compare the Predicted Distance vs Velocity to Manufacturer's Data

This velocity-distance data is provided by the manufacturer of the [308 Winchester 185gr Juggernaut Target]() bullet.

```
> distance := convert([0, 100, 200, 300, 400], units, 'yd', 'm') :
  velocity := convert([2628, 2470, 2318, 2172, 2030], units, 'ft/s', 'm/s') :
```

```
> manu_dist_vs_vel := plot(distance, velocity, style = point, symbol = solidcircle, symbolsize = 15,
    color = black, legend = "Manufacturer's Data") :
  theory_dist_vs_vel := plots:-odeplot(res, [x(t), v(t)], t = 0 ..0.55, axes = boxed, legend
    = "Solution of Differential Equations", labels = ["Horizontal Distance (m)", "Velocity (m/s)"],
    plot_style) :

  plots:-display(manu_dist_vs_vel, theory_dist_vs_vel, labels = ["Horizontal Distance (m)",
    "Velocity (m/s)"], title = "Bullet Velocity vs Distance: Model vs Manufacturer's Data",
    plot_style)
```



A good agreement is found, giving confidence in the theoretical model.